



# **Bayesian Estimation of GPM with DYNARE**

Michel Juillard

---



## Outline

### Bayesian Estimation of GPM with Dynare

Macro–Financial Linkages, Oil Prices and Deflation IMF  
workshop

Michel Juillard

January 6, 2009

1. Introduction to Bayesian estimation
2. Bayesian estimation in Dynare
3. Dealing with nonstationary variables
4. Example: A simple GPM model for the US
5. Dynare macro language
6. Example: A 6–country GPM model



## Introduction to Bayesian estimation

- ▶ Uncertainty and *a priori* knowledge about the model and its parameters are described by prior probabilities
- ▶ Confrontation to the data leads to a revision of these probabilities (posterior probabilities)
- ▶ Point estimates are obtained by minimizing a loss function (analogous to economic decision under uncertainty)
- ▶ Testing and model comparison is done by comparing posterior probabilities



## Bayesian ingredients

- ▶ Choosing prior density
- ▶ Computing posterior mode
- ▶ Simulating posterior distribution
- ▶ Computing point estimates and confidence regions
- ▶ Computing posterior probabilities



## Prior density

$$p(\theta_A|A)$$

where  $A$  represents the model and  $\theta_A$ , the parameters of that model.

The prior density describes *a priori* beliefs, before considering the data.



## Likelihood function

- ▶ Conditional density

$$p(\mathbf{y}|\theta_A, A)$$

- ▶ Conditional density for dynamic timeseries models

$$p(\mathbf{Y}_T|\theta_A, A) = p(y_0|\theta_A, A) \prod_{t=1}^T p(y_t|\mathbf{Y}_{t-1}, \theta_A, A)$$

where  $\mathbf{Y}_T$  are the observations until period  $T$

- ▶ Likelihood function

$$\mathcal{L}(\theta_A|\mathbf{Y}_T, A) = p(\mathbf{Y}_T|\theta_A, A)$$



## Marginal density

$$\begin{aligned} p(\mathbf{y}|A) &= \int_{\Theta_A} p(\mathbf{y}, \theta_A|A) d\theta_A \\ &= \int_{\Theta_A} p(\mathbf{y}|\theta_A, A) p(\theta_A|A) d\theta_A \end{aligned}$$



## Posterior density

- ▶ Posterior density

$$p(\theta_A|\mathbf{Y}_T, A) = \frac{p(\theta_A|A)p(\mathbf{Y}_T|\theta_A, A)}{p(\mathbf{Y}_T|A)}$$

- ▶ Unnormalized posterior density or posterior density kernel

$$p(\theta_A|\mathbf{Y}_T, A) \propto p(\theta_A|A)p(\mathbf{Y}_T|\theta_A, A)$$



$$\begin{aligned} p(\tilde{\mathbf{Y}}|\mathbf{Y}_T, A) &= \int_{\Theta_A} p(\tilde{\mathbf{Y}}, \theta_A | \mathbf{Y}_T, A) d\theta_A \\ &= \int_{\Theta_A} p(\tilde{\mathbf{Y}}|\theta_A, \mathbf{Y}_T, A) p(\theta_A | \mathbf{Y}_T, A) d\theta_A \end{aligned}$$

$$\begin{aligned} R(a) &= E[L(a, \theta)] \\ &= \int_{\Theta_A} L(a, \theta_A) p(\theta_A) d\theta_A \end{aligned}$$

where  $L(a, \theta)$  is the loss function associated with decision  $a$  when parameters take value  $\theta_A$ .



Action: deciding that the estimated value of  $\theta_A$  is  $\tilde{\theta}_A$

- ▶ Point estimate:

$$\hat{\theta}_A = \arg \min_{\tilde{\theta}_A} \int_{\Theta_A} L(\tilde{\theta}_A, \theta_A) p(\theta_A | \mathbf{Y}_T, A) d\theta_A$$

- ▶ Quadratic loss function:

$$\hat{\theta}_A = E(\theta_A | \mathbf{Y}_T, A)$$

- ▶ Zero-one loss function:  $\hat{\theta}_A =$  posterior mode



$$P(\theta \in C) = \int_C p(\theta) d\theta = 1 - \alpha$$

is a  $100(1 - \alpha)\%$  credible set for  $\theta$  with respect to  $p(\theta)$ .

A  $100(1 - \alpha)\%$  highest probability density (HPD) credible set for  $\theta$  with respect to  $p(\theta)$  is a  $100(1 - \alpha)\%$  credible set with the property

$$p(\theta_1) \geq p(\theta_2) \quad \forall \theta_1 \in C \text{ and } \forall \theta_2 \in \bar{C}$$



$$E(h(\theta_A)) = \int_{\Theta_A} h(\theta_A) p(\theta_A | \mathbf{Y}_T, A) d\theta_A$$

$$\approx \frac{1}{N} \sum_{k=1}^N h(\theta_A^k)$$

where  $\theta_A^k$  is drawn from  $p(\theta_A | \mathbf{Y}_T, A)$ .

1. Draw a starting point  $\theta^\circ$  which  $p(\theta) > 0$  from a starting distribution  $p^\circ(\theta)$ .



## Metropolis algorithm (continued)

2. For  $t = 1, 2, \dots$

1. Draw a *proposal*  $\theta^*$  from a *jumping* distribution

$$J(\theta^* | \theta^{t-1}) = N(\theta^{t-1}, c\Sigma_{\text{mode}})$$

2. Compute the acceptance ratio

$$r = \frac{p(\theta^*)}{p(\theta^{t-1})}$$

3. Set

$$\theta^t = \begin{cases} \theta^* & \text{with probability } \min(r, 1) \\ \theta^{t-1} & \text{otherwise.} \end{cases}$$

## In practice ...

- ▶ fix scale factor  $c$  so as to obtain a 25% average acceptance ratio
- ▶ discard first 50% of the draws



# Potential Scale Reduction Factor

If we have simulated  $m$  independent sequences of  $n$  draws, a particular draw of scalar  $\theta$  is noted  $\theta_{ij}$  with  $i = 1, \dots, n$  and  $j = 1, \dots, m$ .

$$\begin{aligned}
 B &= \frac{n}{m-1} \sum_{j=1}^m (\bar{\theta}_{.j} - \bar{\theta}_{..})^2 \\
 W &= \frac{1}{m} \sum_{j=1}^m \frac{1}{n-1} \sum_{i=1}^n (\theta_{ij} - \theta_{.j})^2 \\
 \widehat{\text{var}}^+(\theta | \mathbf{Y}_T, A) &= \frac{n-1}{n} W + \frac{1/n}{B} \\
 \hat{R} &= \sqrt{\frac{\widehat{\text{var}}^+(\theta | \mathbf{Y}_T, A)}{W}}
 \end{aligned}$$

# Multivariate PSRF

$$\begin{aligned}
 \hat{V} &= \frac{n-1}{n} W + \left(1 + \frac{1}{m}\right) B/n \\
 W &= \frac{1}{m(n-1)} \sum_{j=1}^m \sum_{i=1}^n (\theta_{ij} - \bar{\theta}_{.j})(\theta_{ij} - \bar{\theta}_{.j})' \\
 B/n &= \frac{1}{m-1} \sum_{j=1}^m (\bar{\theta}_{.j} - \bar{\theta}_{..})(\bar{\theta}_{.j} - \bar{\theta}_{..})' \\
 \hat{R}^p &= \frac{n-1}{n} + \frac{m+1}{m} \lambda_1
 \end{aligned}$$

$\lambda_1$  is the largest eigenvalue of  $W^{-1}B/n$



# Model comparison

The ratio of posterior probabilities of two models is

$$\frac{P(A_j | \mathbf{Y}_T)}{P(A_k | \mathbf{Y}_T)} = \frac{P(A_j)}{P(A_k)} \frac{p(\mathbf{Y}_T | A_j)}{p(\mathbf{Y}_T | A_k)}$$

In favor of the model  $A_j$  versus the model  $A_k$ :

- ▶ the **prior odds ratio** is  $P(A_j)/P(A_k)$
- ▶ the **Bayes factor** is  $p(\mathbf{Y}_T | A_j)/p(\mathbf{Y}_T | A_k)$
- ▶ the **posterior odds ratio** is  $P(A_j | \mathbf{Y}_T)/P(A_k | \mathbf{Y}_T)$

# Laplace approximation

$$\begin{aligned}
 p(\mathbf{Y}_T, A) &= \int_{\theta_A} p(\theta_A | \mathbf{Y}_T, A) p(\theta_A | A) d\theta_A \\
 \hat{p}(\mathbf{Y}_T | A) &= (2\pi)^{\frac{k}{2}} |\Sigma_{\theta^M}|^{-\frac{1}{2}} p(\theta_A^M | \mathbf{Y}_T, A) p(\theta_A^M | A)
 \end{aligned}$$

where  $\theta_A^M$  is the posterior mode.



# Geweke (1999) modified harmonic mean

$$\begin{aligned}
 p(\mathbf{Y}_T|A) &= \int_{\theta_A} p(\theta_A|\mathbf{Y}_T, A)p(\theta_A|A)d\theta_A \\
 \hat{p}(\mathbf{Y}_T|A) &= \left[ \frac{1}{n} \sum_{i=1}^n \frac{f(\theta_A^{(i)})}{p(\theta_A^{(i)}|\mathbf{Y}_T, A)p(\theta_A^{(i)}|A)} \right]^{-1} \\
 f(\theta) &= p^{-1}(2\pi)^{\frac{k}{2}}|\Sigma_{\theta}|^{-\frac{1}{2}} \exp \left\{ -\frac{1}{2}(\theta - \bar{\theta})'\Sigma_{\theta}^{-1}(\theta - \bar{\theta}) \right\} \\
 &\quad \times \left\{ (\theta - \bar{\theta})'\Sigma_{\theta}^{-1}(\theta - \bar{\theta}) \leq F_{\chi_k^2(p)}^{-1} \right\}
 \end{aligned}$$

Bayesian estimation in Dynare

with  $p$  an arbitrary probability and  $k$ , the number of estimated parameters.



## Priors in DYNARE

NORMAL_PDF	$N(\mu, \sigma)$	$R$
GAMMA_PDF	$G_2(\mu, \sigma, p_3)$	$[p_3, +\infty)$
BETA_PDF	$B(\mu, \sigma, p_3, p_4)$	$[p_3, p_4]$
INV_GAMMA_PDF	$IG_1(\mu, \sigma)$	$R^+$
UNIFORM_PDF	$U(p_3, p_4)$	$[p_3, p_4]$

By default,  $p_3 = 0, p_4 = 1$ .

## How to choose priors

- ▶ the shape should be consistent with the domain of definition of the parameter
- ▶ use values obtained in other studies (micro or macro)
- ▶ check the graph of the priors
- ▶ check the implication of your priors by running `stoch_simul` with parameters set at prior mean
- ▶ compare moments of endogenous variables in previous simulation with empirical moments of observed variables
- ▶ do sensitivity tests by widening your priors





# Estimation strategy

- ▶ After (log-)linearization around the deterministic steady state, the linear rational expectation model needs to be solved (AIM, Kind and Watson, Klein, Sims)
- ▶ The model can then be written in state space form
- ▶ It is an unobserved component model
- ▶ Its likelihood is computed via the Kalman filter
- ▶ These steps are common to Maximum Likelihood estimation or a Bayesian approach



# State space representation (I)

After solution of a first order approximation of a DSGE model, we obtain a linear dynamic model of the form

$$y_t = \bar{y} + g_y \hat{y}_{t-1}^s + g_u u_t$$

the vector  $\hat{y}_{t-1}^s$  contains the endogenous state variables, the predetermined variables among  $y_t$ , with as many lags as required by the dynamic of the model.



# State space representation (II)

The transition equation describes the dynamics of the state variables:

$$\hat{y}_t^{(1)} = g_y^{(1)} \hat{y}_{t-1}^{(1)} + g_u^{(1)} u_t$$

where  $g_x^{(1)}$  and  $g_u^{(1)}$  are the appropriate submatrices of  $g_x$  and  $g_u$ , respectively.  $y_t^{(1)}$  is the union of the state variables  $y_t^s$ , including all necessary lags, and  $y_t^*$ , the observed variables. The  $g_y^{(1)}$  matrix can have eigenvalues equal to one.



# Other variables

The variables that are neither predetermined nor observed,  $y_t^{(2)}$ , play no role in the estimation of the parameters, and their filtered or smoothed values can be recovered from the filtered or smoothed values of  $\hat{y}_t^{(1)}$  thanks to the following relationship:

$$\hat{y}_t^{(2)} = g_x^{(2)} \hat{y}_{t-1}^{(1)} + g_u^{(2)} u_t$$



We consider measurement equations of the type

$$y_t^* = \bar{y} + M\hat{y}_t^{(1)} + x_t + \epsilon_t$$

where  $M$  is the selection matrix that recovers  $\hat{y}_t^*$  out of  $\hat{y}_t^{(1)}$ ,  $x_t$  is a deterministic component<sup>1</sup> and  $\epsilon_t$  is a vector of measurement errors.

In addition, we have, the two following covariance matrices:

$$E(u_t u_t') = Q$$

$$E(\epsilon_t \epsilon_t') = H$$

---

<sup>1</sup>Currently, Dynare only accomodates linear trends

## Unit root processes

Dealing with nonstationary variables

- ▶ find a natural representation in the state space form
- ▶ the deterministic components of random walk with drift is better included in the measurement equation

## Initialization of the Kalman filter

- ▶ stationary variables: unconditional mean and variance
- ▶ nonstationary variables: initial point is an additional parameter of the model (De Jong), arbitrary initial point and infinite variance (Durbin and Koopman).
- ▶ Durbin and Koopman strategy: compute the limit of the Kalman filter equations when initial variance tends toward infinity.
- ▶ Problem with cointegrated models.



## The Schur decomposition of the transition matrix

In the transition equation

$$\hat{y}_t^{(1)} = g_x^{(1)} \hat{y}_{t-1}^{(1)} + g_u^{(1)} u_t$$

we propose to perform a reordered real Schur decomposition on transition matrix  $g_x$ :

$$g_x^{(1)} = W \begin{bmatrix} T_{11} & T_{12} \\ 0 & T_{22} \end{bmatrix} W'$$

where  $T_{11}$  and  $T_{22}$  are quasi upper-triangular matrices and  $W$  is an orthogonal matrix. The reordering is such that the absolute value of the eigenvalues of  $T_{11}$  are all equal to 1 while the eigenvalues of  $T_{22}$  are all smaller than 1 in modulus. When there are cointegrating relationships between the state variables, there are obviously less unit roots in the system than the number of nonstationary variables in the model. The dimension of  $T_{11}$  reflects this fact.



## New state space formulation

It is then natural to rewrite the transition equation in transformed variables as

$$W' \hat{y}_t^{(1)} = T W' \hat{y}_{t-1}^{(1)} + W' g_u u_t$$

and the measurement equation as

$$\tilde{y}_t^* = M W' \hat{y}_t^{(1)} + \epsilon_t$$

Note that in this formulation of the state space representation, only the state variables are transformed, structural shocks and measurement errors stay the same as in the original formulation.



## New notations

In what follows, we write the state space model as

$$\begin{aligned} y_t &= Z a_t + \epsilon_t \\ a_t &= T a_{t-1} + R \eta_t \\ E(\epsilon_t \epsilon_t') &= H \\ E(\eta_t \eta_t') &= Q \end{aligned}$$



# Equivalence in notation

$$\begin{aligned}
 y_t &= \tilde{y}_t^* \\
 Z &= MW \\
 a_t &= W' \hat{y}^{(1)} \\
 WTW' &= g_x \\
 R &= W' g_u \\
 \eta_t &= u_t
 \end{aligned}$$

# Diffuse initialization of the filter

The initial values for the state variables are  $a_0 = 0$ . This is the unconditional mean of the stationary elements in  $a_t$  and has no effects for the nonstationary ones.

Following Durbin and Koopman, we set

$$\begin{aligned}
 P_0 &= P_0^\infty + P_0^* \\
 &= \begin{bmatrix} I & 0 \\ 0 & 0 \end{bmatrix} + \begin{bmatrix} 0 & 0 \\ 0 & \Sigma_{\tilde{a}} \end{bmatrix}
 \end{aligned}$$

where  $I$  is an identity matrix of the same dimensions as  $T_{11}$ . It corresponds to the diffuse prior on the initial values of the stochastic trends.  $\Sigma_{\tilde{a}}$  is the covariance matrix of the stationary part of  $a_t$ .

# Computation of $\Sigma_{\tilde{a}}$

$\Sigma_{\tilde{a}}$  is the covariance matrix of  $\tilde{a}_t$  with dynamics

$$\tilde{a}_t = T_{12} a_{t-1} + \tilde{R} \eta_t$$

or

$$\Sigma_{\tilde{a}} = T_{12} \Sigma_{\tilde{a}} T_{12}' + \tilde{R} Q \tilde{R}'$$

where  $\tilde{R}$  is the conforming submatrix of  $R$ . As  $T_{12}$  is already quasi upper-triangular, it is only necessary to use part of the usual algorithm for the Lyapunov equation.

# The diffuse step

While  $P_t^\infty$  is different from zero, the filter (and smoother) is in a diffuse step. When  $t > d$ , the procedure falls back on standard recursions.

At  $t = 0$

$$E(a_{1|0}) = P_{1|0} = P_{1|0}^\infty + P_{1|0}^*$$

# Recursion

# Log-likelihood

$$\begin{aligned}
 F_t^\infty &= ZP_{t|t-1}^\infty Z' \\
 F_t^* &= ZP_t^* Z' + H \\
 K_t^\infty &= TP_{t|t-1}^\infty Z' (F_t^\infty)^{-1} \\
 K_t^* &= T \left( P_{t|t-1}^* Z' (F_t^\infty)^{-1} - P_{t|t-1}^\infty Z' (F_t^\infty)^{-1} F_t^* (F_t^\infty)^{-1} \right) \\
 v_t &= y_t - Za_{t|t-1} \\
 a_{t+1|t} &= Ta_{t|t-1} + K_t^\infty v_t \\
 P_{t+1|t}^\infty &= TP_{t|t-1}^\infty (T' - Z' K_t^{\infty'}) \\
 P_{t+1|t}^* &= -TP_{t|t-1}^\infty Z' K_t^{*'} + TP_{t|t-1}^* (T' - Z' K_t^{\infty'}) + RQR'
 \end{aligned}$$

where  $a_{t|t-1} = E_{t-1} a_t$ .

The log-likelihood is given by

$$-\frac{nT}{2} \ln 2\pi - \frac{1}{2} \sum_{t=1}^T \ln |F_t| - \frac{1}{2} \sum_{t=1}^T v_t' F_t^{-1} v_t$$



# A simple GPM model for the US

Example: A simple GPM model for the US

```

var
GROWTH_US GROWTH4_US GROWTH_BAR_US GROWTH4_BAR_US
RS_US DRS_US RR_US RR_BAR_US
PIETAR_US PIE_US PIE4_US LCPI_US E1_PIE_US E4_PIE4_US
LGDP_US LGDP_BAR_US G_US Y_US E1_Y_US
UNR_US UNR_GAP_US UNR_BAR_US UNR_G_US
E_US E2_US BLT_US BLT_BAR_US;

varexo
RES_RS_US RES_RR_BAR_US RES_PIETAR_US RES_PIE_US RES_G_US RES_LGDP_BAR_US
RES_Y_US RES_UNR_GAP_US RES_UNR_BAR_US RES_UNR_G_US
RES_BLT_US RES_BLT_BAR_US;

parameters
gamma1_US gamma2_US gamma3_US gamma4_US
lambda1_RS_US rho_US rr_bar_US_ss
lambda1_US lambda2_US lambda3_US
tau_US growth_US_ss
beta1_US beta2_US beta3_US beta_fact_US beta_reergap_US
alpha1_US alpha2_US alpha3_US
kappa_US theta_US;
    
```



(continued)

```
beta_fact_US=0.0241;
beta_reergap_US=0.0423;
alpha1_US=0.8235;
alpha2_US=0.1823;
alpha3_US=0.3649;
beta1_US=0.6549;
beta2_US=0.0694;
beta3_US=0.1866;
gamma1_US=0.7107;
gamma2_US=0.9104;
gamma4_US=0.2052;
growth_US_ss=2.2729;
kappa_US=20.0773;
lambda1_US=0.848;
lambda1_RS_US=0;
lambda2_US=0.1801;
lambda3_US=0.0707;
pietar_US_ss=2.5;
rho_US=0.2901;
rr_bar_US_ss=1.7285;
tau_US=0.0274;
theta_US=1.0708;
```

⏪ ⏩ ⏴ ⏵ ⏶ ⏷ ⏸ ⏹ ⏺ ⏻ ⏼ ⏽ ⏾ ⏿ 🔍 ↻

(continued)

```
model(linear);
GROWTH_US = 4*(LGDP_US-LGDP_US(-1)) ;
GROWTH4_US = LGDP_US-LGDP_US(-4) ;
GROWTH_BAR_US = 4*(LGDP_BAR_US-LGDP_BAR_US(-1)) ;
GROWTH4_BAR_US = LGDP_BAR_US-LGDP_BAR_US(-4) ;
RS_US = gamma1_US*RS_US(-1)+(1-gamma1_US)*(RR_BAR_US+PIE4_US(+3)
+gamma2_US*(PIE4_US(+3)-PIETAR_US)+gamma4_US*Y_US)+RES_RS_US ;
DRS_US = RS_US-RS_US(-1) ;
RR_US = RS_US-PIE_US(+1) ;
RR_BAR_US = rho_US*rr_bar_US_ss+(1-rho_US)*RR_BAR_US(-1)+RES_RR_BAR_US ;
PIETAR_US = PIETAR_US(-1)-RES_PIETAR_US ;
PIE_US = lambda1_US*PIE4_US(+4)+(1-lambda1_US)*PIE4_US(-1)
+lambda2_US*Y_US(-1)-RES_PIE_US ;
LCPI_US = LCPI_US(-1)+PIE_US/4 ;
PIE4_US = (PIE_US+PIE_US(-1)+PIE_US(-2)+PIE_US(-3))/4 ;
E4_PIE4_US = PIE4_US(+4) ;
E1_PIE_US = PIE_US(+1) ;
LGDP_BAR_US = LGDP_BAR_US(-1)+G_US/4+RES_LGDP_BAR_US ;
G_US = tau_US*growth_US_ss+(1-tau_US)*G_US(-1)+RES_G_US ;
E1_Y_US = Y_US(+1) ;
Y_US = LGDP_US-LGDP_BAR_US ;
```

⏪ ⏩ ⏴ ⏵ ⏶ ⏷ ⏸ ⏹ ⏺ ⏻ ⏼ ⏽ ⏾ ⏿ 🔍 ↻

(continued)

```
UNR_GAP_US = alpha1_US*UNR_GAP_US(-1)+alpha2_US*Y_US+RES_UNR_GAP_US ;
UNR_GAP_US = UNR_BAR_US-UNR_US ;
UNR_BAR_US = UNR_BAR_US(-1)+UNR_G_US+RES_UNR_BAR_US ;
UNR_G_US = (1-alpha3_US)*UNR_G_US(-1)+RES_UNR_G_US ;

E_US = -RES_BLT_US ;
BLT_US = BLT_BAR_US-kappa_US*Y_US(+4)-RES_BLT_US ;
BLT_BAR_US = BLT_BAR_US(-1)+RES_BLT_BAR_US ;
E2_US = theta_US*(0.04*(E_US(-1)+E_US(-9))+0.08*(E_US(-2)+E_US(-8))
+0.12*(E_US(-3)+E_US(-7))+0.16*(E_US(-4)+E_US(-6))+0.2*E_US(-5)) ;
Y_US = beta1_US*Y_US(-1)+beta2_US*Y_US(+1)-beta3_US*(RR_US(-1)-RR_BAR_US(-1))
+beta_fact_US*FACT_US+beta_reergap_US*(REER_T_US(-1)-REER_T_BAR_US(-1))
-theta_US*(0.04*(E_US(-1)+E_US(-9))+0.08*(E_US(-2)+E_US(-8))
+0.12*(E_US(-3)+E_US(-7))+0.16*(E_US(-4)+E_US(-6))+0.2*E_US(-5))+RES_Y_US ;

end;
```

⏪ ⏩ ⏴ ⏵ ⏶ ⏷ ⏸ ⏹ ⏺ ⏻ ⏼ ⏽ ⏾ ⏿ 🔍 ↻

(continued)

```
shocks;
var RES_RS_US; stderr 0.7;
var RES_RR_BAR_US; stderr 0.2;
var RES_PIETAR_US; stderr 0;
var RES_PIE_US; stderr 0.7;
var RES_G_US; stderr 0.1;
var RES_LGDP_BAR_US; stderr 0.1;
var RES_Y_US; stderr 0.25;
var RES_UNR_GAP_US; stderr 0.2;
var RES_UNR_BAR_US; stderr 0.1;
var RES_UNR_G_US; stderr 0.1;
var RES_BLT_US; stderr 0.1;
var RES_BLT_BAR_US; stderr 0.1;
end;

check;

stoch_simul;
```

⏪ ⏩ ⏴ ⏵ ⏶ ⏷ ⏸ ⏹ ⏺ ⏻ ⏼ ⏽ ⏾ ⏿ 🔍 ↻

(continued)

```

estimated_params;
alpha1_US,beta_pdf,0.8,0.1;
alpha2_US,gamma_pdf,0.3,0.2;
alpha3_US,beta_pdf,0.5,0.2;
beta1_US,gamma_pdf,0.75,0.1;
beta2_US,beta_pdf,0.15,0.1;
beta3_US,gamma_pdf,0.2,0.0500;
gamma1_US,beta_pdf,0.5,0.0500;
gamma2_US,gamma_pdf,1.5,0.3000;
gamma4_US,gamma_pdf,0.2,0.0500;
growth_US_ss,normal_pdf,2.5,0.2500;
kappa_US,gamma_pdf,20.000,0.5000;
lambda1_US,beta_pdf,0.5,0.1;
lambda2_US,gamma_pdf,0.25,0.05;
lambda3_US,gamma_pdf,0.120,0.0500;
rho_US,beta_pdf,0.9,0.05;
rr_bar_US_ss,normal_pdf,2.000,0.3000;
tau_US,beta_pdf,0.1,0.05;
theta_US,gamma_pdf,1.000,0.5000;

```



(continued)

```

stderr RES_BLT_BAR_US,inv_gamma_pdf,0.200,Inf;
stderr RES_BLT_US,inv_gamma_pdf,0.400,Inf;
stderr RES_G_US,inv_gamma_pdf,0.100,Inf;
stderr RES_LGDP_BAR_US,inv_gamma_pdf,0.1,Inf;
stderr RES_PIE_US,inv_gamma_pdf,0.700,Inf;
stderr RES_RR_BAR_US,inv_gamma_pdf,0.200,Inf;
stderr RES_RS_US,inv_gamma_pdf,0.700,Inf;
stderr RES_UNR_BAR_US,inv_gamma_pdf,0.100,Inf;
stderr RES_UNR_G_US,inv_gamma_pdf,0.100,Inf;
stderr RES_UNR_GAP_US,inv_gamma_pdf,0.200,Inf;
stderr RES_Y_US,inv_gamma_pdf,0.250,Inf;
corr RES_BLT_US,RES_G_US,beta_pdf,0.650,0.0500;
corr RES_LGDP_BAR_US,RES_PIE_US,beta_pdf,0.100,0.0300;
end;

```



(continued)

```

varobs UNR_US LGDP_US LCPI_US RS_US BLT_US;

observation_trends;
LGDP_US (growth_US_ss/4);
end;

estimation(datafile=data6ctryCORE94,mh_replic=0,diffuse_filter);

```



### Dynare macro language



## Macro language

It extends the language of MOD files by adding "macro" commands for doing the following tasks: source file inclusion, replicating blocks of equations through loops, conditional inclusion of code...

Technically, this macro language is totally independent of the basic Dynare language, and is processed by a separate component of the Dynare pre-processor. The macro processor transforms a MOD file with macros into a MOD file without macros (doing expansions/inclusions), and then feeds it to the Dynare parser. The advantage of such a design choice is to clearly separate the macro language from the rest of the language, which gives a simpler language semantics and a simpler code.



## Directives

All directives begin with an at-sign followed by a pound sign (@#) and occupy exactly one line. However, a directive can be continued on next line by adding two anti-slashes (\ \) at the end of the line to be continued.

A directive produces no output, but serves to give instructions to the macro processor.



## Variables

The macro processor maintains its own list of variables.

Variables can be of four types:

- ▶ integer
- ▶ string
- ▶ array of integers
- ▶ array of strings



## Expressions

It is possible to construct expressions, using the following operators:

- ▶ on integers:
  - ▶ arithmetic operators (,-,\*,/,+)
  - ▶ comparison operators (<, >, <=, >=, ==, !=)
  - ▶ logical operators (&&, ||, !)
  - ▶ inclusion operator (in)
- ▶ on strings:
  - ▶ comparison operators (==, !=)
  - ▶ inclusion operator (in)
  - ▶ concatenation (+)
  - ▶ extraction of substrings (if  $s$  is a string, then one can write  $s[3]$  or  $s[4:6]$ )





## Expressions (continued)

- ▶ on arrays:
  - ▶ dereferencing (if `v` is an array, then `v[2]` is its 2nd element)
  - ▶ concatenation (+)
  - ▶ difference (-): returns the first operand from which the elements of the second operand have been removed
  - ▶ extraction of sub-arrays (with `v[4:6]`)
  - ▶ shortcut for integer ranges (`1:5` is equivalent to `[1,2,3,4,5]`)

Expressions can be used at two places:

- ▶ inside macro directives, directly
- ▶ outside macro directives, between an at-sign and curly braces, like: `@{expr}`. The macro processor will substitute the expression with its value



## Define directive

The value of a variable can be defined with the `@#define` directive.

Isolated examples:

```
@#define x = 5
@#define y = "foo"
@#define v = [ 1, 2, 4 ]
@#define w = [ "foo", "bar" ]
@#define z = 3+v[2]
```



## Integrated example

```
@#define x = ["B", "C"]
@#define i = 1
```

```
model;
  A = @{x[i]};
end;
```

Is equivalent to:

```
model;
  A = B;
end;
```



## Inclusion directive

This directive simply includes the content of another file at the place where it is inserted.

```
@#include "modelcomponent.mod"
```

It is possible to include a file from an included file (nested includes).



## Loop directive

Loops are constructed with the following syntax

```
model;
@#for country in ["home", "foreign"]
    GDP_{country} = K_{country}^a * L_{country}^(1-a)
@#endfor
end;
```

Is equivalent to:

```
model;
    GDP_home = K_home^a * L_home^(1-a);
    GDP_foreign = K_foreign^a * L_foreign^(1-a);
end;
```



## Conditional inclusion directives

The syntax is either:

```
@#if integer_expression
...body if expression = 1...
@#endif
```

or:

```
@#if integer_expression
...body if expression = 1...
@#else
...body if expression = 0...
@#endif
```



## Echo and error directives

It is possible to ask the macro processor to display a message on standard output:

```
@#echo "message"
```

It is also possible to ask the macro processor to fail with a message (only useful inside a conditional inclusion directive).

```
@#error "message"
```



## Saving the macro-expanded MOD file

It is possible to save the output of macro-expansion, using the `savemacro` option on the Dynare command line. It can be useful for debugging purposes.

If MOD file is `filename.mod`, then the macro-expanded version will be saved in `filename-macroexp.mod`.



# A 6-country GPM model

## Example: A 6-country GPM model

```
    /*** list of countries
    @#define countries = {"EA", "EU", "JA", "LA", "RC", "US"}

    /*** variables and parameters declarations
    @#for c in countries
    var
    GROWTH_@{c} GROWTH4_@{c} GROWTH_BAR_@{c} GROWTH4_BAR_@{c}
    RS_@{c} DRS_@{c} RR_@{c} RR_BAR_@{c} RESN_RS_@{c} DOT_REER_M_BAR_@{c}
    PIETAR_@{c} PIE_@{c} PIE4_@{c} LCPI_@{c} E1_PIE_@{c} E4_PIE4_@{c}
    LGDP_@{c} LGDP_BAR_@{c} G_@{c} Y_@{c} E1_Y_@{c}
    REER_M_@{c} REER_M_BAR_@{c} REER_T_@{c} REER_T_BAR_@{c} FACT_@{c};

    varexo
    RES_RS_@{c} RES_RR_BAR_@{c} RES_PIETAR_@{c} RES_PIE_@{c} RES_G_@{c} RES_LGDP_BAR_@{c}
    RES_Y_@{c};

    parameters
    gammal_@{c} gamma2_@{c} gamma3_@{c} gamma4_@{c}
    lambdal_RS_@{c} rho_@{c} rr_bar_@{c}_ss
    lambdal_@{c} lambda2_@{c} lambda3_@{c}
    tau_@{c} growth_@{c}_ss
    beta1_@{c} beta2_@{c} beta3_@{c} beta_fact_@{c} beta_reergap_@{c};
```

◀ ▶ ⏪ ⏩ ⏴ ⏵ ⏶ ⏷ ⏸ ⏹ ⏺ ⏻ ⏼ ⏽ ⏾ ⏿ 🔍

◀ ▶ ⏪ ⏩ ⏴ ⏵ ⏶ ⏷ ⏸ ⏹ ⏺ ⏻ ⏼ ⏽ ⏾ ⏿ 🔍

(continued)

```
@#if c == "EU" || c == "JA" || c == "US"
    var UNR_@{c} UNR_GAP_@{c} UNR_BAR_@{c} UNR_G_@{c};
    varexo RES_UNR_GAP_@{c} RES_UNR_BAR_@{c} RES_UNR_G_@{c};
    parameters alpha1_@{c} alpha2_@{c} alpha3_@{c};
    @#endif

@#if c == "US"
    var E_@{c} E2_@{c} BLT_@{c} BLT_BAR_@{c};
    varexo RES_BLT_@{c} RES_BLT_BAR_@{c};
    parameters kappa_@{c} theta_@{c};
    @#endif

@#if c != "US"
    var LS_@{c} LZ_@{c} LZ_BAR_@{c} DOT_LZ_BAR_@{c} LZ_E_@{c} LZ_GAP_@{c};
    varexo RES_RR_DIFF_@{c} RES_LZ_BAR_@{c} RES_DOT_LZ_BAR_@{c};
    parameters chi_@{c} phi_@{c} dot_lz_bar_@{c}_ss;
    @#endif
```

◀ ▶ ⏪ ⏩ ⏴ ⏵ ⏶ ⏷ ⏸ ⏹ ⏺ ⏻ ⏼ ⏽ ⏾ ⏿ 🔍

(continued)

```
    parameters
    @#for c1 in countries
        @#if c1 != c
            imp_@{c}_@{c1} trade_@{c}_@{c1} exp_@{c}_@{c1}
        @#endif
    @#endfor
    ;

@#if "EA" in countries
    @#include "parameter_values_EA.mod"
    @#endif

@#if "EU" in countries
    @#include "parameter_values_EU.mod"
    @#endif

@#if "JA" in countries
    @#include "parameter_values_JA.mod"
    @#endif
```

◀ ▶ ⏪ ⏩ ⏴ ⏵ ⏶ ⏷ ⏸ ⏹ ⏺ ⏻ ⏼ ⏽ ⏾ ⏿ 🔍

(continued)

```

@if "LA" in countries
  @include "parameter_values_LA.mod"
@endif

@if "RC" in countries
  @include "parameter_values_RC.mod"
@endif

@if "US" in countries
  @include "parameter_values_US.mod"
@endif

```



(continued)

```

model(linear);
@#for c in countries
  GROWTH_@{c} = 4*(LGDP_@{c}-LGDP_@{c}(-1)) ;
  GROWTH4_@{c} = LGDP_@{c}-LGDP_@{c}(-4) ;
  GROWTH_BAR_@{c} = 4*(LGDP_BAR_@{c}-LGDP_BAR_@{c}(-1)) ;
  GROWTH4_BAR_@{c} = LGDP_BAR_@{c}-LGDP_BAR_@{c}(-4) ;
  RS_@{c} = gamma1_@{c}*RS_@{c}(-1)+(1-gamma1_@{c})*(RR_BAR_@{c}+PIE4_@{c}(+3)
    +gamma2_@{c}*(PIE4_@{c}(+3)-PIETAR_@{c}))+gamma4_@{c}*Y_@{c}))+RESN_RS_@{c} ;
  RESN_RS_@{c} = lambda1_RS_@{c}*RESN_RS_@{c}(-1)+RESN_RS_@{c} ;
  DRS_@{c} = RS_@{c}-RS_@{c}(-1) ;
  RR_@{c} = RS_@{c}-PIE_@{c}(+1) ;
  RR_BAR_@{c} = rho_@{c}*rr_bar_@{c}_ss+(1-rho_@{c})*RR_BAR_@{c}(-1)+RES_RR_BAR_@{c} ;
  PIETAR_@{c} = PIETAR_@{c}(-1)-RES_PIETAR_@{c} ;
  PIE_@{c} = lambda1_@{c}*PIE4_@{c}(+4)+(1-lambda1_@{c})*PIE4_@{c}(-1)
    +lambda2_@{c}*Y_@{c}(-1)+lambda3_@{c}*(REER_M_@{c}-REER_M_@{c}(-1)
    -DOT_REER_M_BAR_@{c}/4)-RES_PIE_@{c} ;
  LCPI_@{c} = LCPI_@{c}(-1)+PIE_@{c}/4 ;
  PIE4_@{c} = (PIE_@{c}+PIE_@{c}(-1)+PIE_@{c}(-2)+PIE_@{c}(-3))/4 ;
  E4_PIE4_@{c} = PIE4_@{c}(+4) ;
  E1_PIE_@{c} = PIE_@{c}(+1) ;

```



(continued)

```

LGDP_BAR_@{c} = LGDP_BAR_@{c}(-1)+G_@{c}/4+RES_LGDP_BAR_@{c} ;
G_@{c} = tau_@{c}*growth_@{c}_ss+(1-tau_@{c})*G_@{c}(-1)+RES_G_@{c} ;
E1_Y_@{c} = Y_@{c}(+1) ;
Y_@{c} = LGDP_@{c}-LGDP_BAR_@{c} ;
DOT_REER_M_BAR_@{c} = 4*(REER_M_BAR_@{c}-REER_M_BAR_@{c}(-1)) ;

@if c == "EU" || c == "JA" || c == "US"
  UNR_GAP_@{c} = alpha1_@{c}*UNR_GAP_@{c}(-1)+alpha2_@{c}*Y_@{c}+RES_UNR_GAP_@{c} ;
  UNR_GAP_@{c} = UNR_BAR_@{c}-UNR_@{c} ;
  UNR_BAR_@{c} = UNR_BAR_@{c}(-1)+UNR_G_@{c}+RES_UNR_BAR_@{c} ;
  UNR_G_@{c} = (1-alpha3_@{c})*UNR_G_@{c}(-1)+RES_UNR_G_@{c} ;
@endif

@if c == "US"
  E_@{c} = -RES_BLT_@{c} ;
  BLT_@{c} = BLT_BAR_@{c}-kappa_@{c}*Y_@{c}(+4)-RES_BLT_@{c} ;
  BLT_BAR_@{c} = BLT_BAR_@{c}(-1)+RES_BLT_BAR_@{c} ;
  E2_@{c} = theta_@{c}*(0.04*(E_@{c}(-1)+E_@{c}(-9))+0.08*(E_@{c}(-2)+E_@{c}(-8))
    +0.12*(E_@{c}(-3)+E_@{c}(-7))+0.16*(E_@{c}(-4)+E_@{c}(-6))+0.2*(E_@{c}(-5)) ;
  Y_@{c} = beta1_@{c}*Y_@{c}(-1)+beta2_@{c}*Y_@{c}(+1)-beta3_@{c}*(RR_@{c}(-1)
    -RR_BAR_@{c}(-1))+beta_fact_@{c}*FACT_@{c}+beta_reergap_@{c}*(REER_T_@{c}(-1)
    -REER_T_BAR_@{c}(-1))-theta_@{c}*(0.04*(E_@{c}(-1)+E_@{c}(-9))
    +0.08*(E_@{c}(-2)+E_@{c}(-8))+0.12*(E_@{c}(-3)+E_@{c}(-7))
    +0.16*(E_@{c}(-4)+E_@{c}(-6))+0.2*(E_@{c}(-5))+RES_Y_@{c} ;
#else
  Y_@{c} = beta1_@{c}*Y_@{c}(-1)+beta2_@{c}*Y_@{c}(+1)-beta3_@{c}*(RR_@{c}(-1)
    -RR_BAR_@{c}(-1))+beta_fact_@{c}*FACT_@{c}+beta_reergap_@{c}*(REER_T_@{c}(-1)
    -REER_T_BAR_@{c}(-1))+RES_Y_@{c} ;
@endif

```



(continued)

```

@if c != "US"
  RR_@{c}-RR_US = 4*(LZ_E_@{c}-LZ_@{c}))+RR_BAR_@{c}-RR_BAR_US-DOT_LZ_BAR_@{c}
    +RES_RR_DIFF_@{c} ;
  LS_@{c} = LZ_@{c}+LCPI_@{c}-LCPI_US ;
  LZ_BAR_@{c} = LZ_BAR_@{c}(-1)+DOT_LZ_BAR_@{c}/4+RES_LZ_BAR_@{c} ;
  DOT_LZ_BAR_@{c} = chi_@{c}*dot_lz_bar_@{c}_ss+(1-chi_@{c})*DOT_LZ_BAR_@{c}(-1)
    +RES_DOT_LZ_BAR_@{c} ;
  LZ_E_@{c} = phi_@{c}*LZ_@{c}(+1)+(1-phi_@{c})*(LZ_@{c}(-1)+2*DOT_LZ_BAR_@{c}/4) ;
  LZ_GAP_@{c} = LZ_@{c}-LZ_BAR_@{c} ;
@endif

REER_M_@{c} =
@#for c1 in countries
  @if c1 != c
    +imp_@{c}_@{c1}*(
      @if c != "US"
        LZ_@{c}
      @endif
      @if c1 != "US"
        -LZ_@{c1}
      @endif
    )
  @endif
@#endfor
;

```



(continued)

```

REER_M_BAR_{c} =
  @#for c1 in countries
    @#if c1 != c
      +imp_{c}_{c1}*(
        @#if c != "US"
          LZ_BAR_{c}
        @#endif
        @#if c1 != "US"
          -LZ_BAR_{c1}
        @#endif
      )
    @#endif
  @#endfor
;
REER_T_{c} =
  @#for c1 in countries
    @#if c1 != c
      +trade_{c}_{c1}*(
        @#if c != "US"
          LZ_{c}
        @#endif
        @#if c1 != "US"
          -LZ_{c1}
        @#endif
      )
    @#endif
  @#endfor
;

```



(continued)

```

REER_T_BAR_{c} =
  @#for c1 in countries
    @#if c1 != c
      +trade_{c}_{c1}*(
        @#if c != "US"
          LZ_BAR_{c}
        @#endif
        @#if c1 != "US"
          -LZ_BAR_{c1}
        @#endif
      )
    @#endif
  @#endfor
;
FACT_{c} =
  @#for c1 in countries
    @#if c1 != c
      +exp_{c}_{c1}*Y_{c1}(-1)
    @#endif
  @#endfor
;
@#endfor
end;

check;

```



(continued)

```

estimated_params;

@#if "EA" in countries
  @#include "estimated_params_EA.mod"
@#endif

@#if "EU" in countries
  @#include "estimated_params_EU.mod"
@#endif

@#if "JA" in countries
  @#include "estimated_params_JA.mod"
@#endif

@#if "LA" in countries
  @#include "estimated_params_LA.mod"
@#endif

@#if "RC" in countries
  @#include "estimated_params_RC.mod"
@#endif

@#if "US" in countries
  @#include "estimated_params_US.mod"
@#endif

end;

```



(continued)

```

@#for c in countries
  varobs RS_{c} LGDP_{c} LCPI_{c};
  @#if c == "EU" || c == "JA" || c == "US"
    varobs UNR_{c};
  @#endif
  @#if c == "US"
    varobs BLT_{c};
  @#endif
@#endfor

observation_trends;

@#for c in countries
  LGDP_{c} (growth_{c}_ss/4);
@#endfor

end;

estimation(datafile=data6ctryCORE94,mh_replic=0,diffuse_filter);

```

